SOUTHWEST FISHERIES CENTER
P.O. BOX 271 LA JOLLA, CA 92038
NATIONAL MARINE FISHERIES SERVICE SOUTHWEST FISHERIES CENTER

APRIL 1987

COMMENTS ON THE SEA LION FOOD-HABITS DATA: SCAT CONTENTS IN THE COLLECTIONS FROM SAN CLEMENTE ISLAND

By

Daniel Goodman

ADMINISTRATIVE REPORT LJ-87-07C

This Administrative Report is issued as an informal document to ensure prompt dissemination of preliminary results, interim reports and special studies. We recommend that it not be abstracted or cited.

COMMENTS ON THE SEA LION FOOD-HABITS DATA: SCAT CONTENTS IN THE COLLECTIONS FROM SAN CLEMENTE ISLAND

Daniel Goodman

National Marine Fisheries Service
Southwest Fisheries Center
8604 La Jolla Shores Drive
La Jolla, California 92038

April 1987

This report was prepared by Daniel Goodman under contract No. 84-JGA-474 for the National Marine Fisheries Service, Southwest Fisheries Center, La Jolla, California. The Statements, findings, conclusions and recommendations herein are those of the author and do not necessarily reflect the views of the National Marine Fisheries Service. Mark S. Lowry of the Southwest Fisheries Center served as Contract Officer's Technical Representative for this contract.

ADMINISTRATIVE REPORT LJ-87-07C

CONTENTS

COMMENTS ON THE SEA LION FOOD-HABITS DATA: SCAT CONTENTS IN THE COLLECTIONS FROM SAN CLEMENTE ISLAND

Daniel Goodman

I. THE DATA SET

The data consist of counts of fish otoliths, shark teeth, and cephalopod beaks, plus observations on presence or absence of red crab shell fragments, in scats collected from a single California sea lion (Zalophus californianus) rookery on San Clemente Island, California. The number of scats on which my analysis was based was 1,347.

Fish otoliths and cephalopod beaks are further identifiable as to left or right, top or bottom, and the maximum number of any one side can be used as a minimum number of prey individuals accounted for. Pelagic red crabs (Pleuroncodes planipes) and sharks can only be scored as present or absent.

The otoliths, teeth and beaks are identifiable to determinate taxonomic units: species, genus or family, as the case may be. Overall, 52 distinct taxa were positively identified--most of these were to the species level. Many, but less than half, of the taxa were observed in only 1 scat in the data set. Many of these taxa were represented by a count of only 1 individual. Thirteen taxa were present in more than 10 scats; 8 taxa were present in more than 100 scats.

The basic sampling unit is the individual scat (i.e. fecal material), which is examined in its entirety, and the prey item counts are ascribed to the scat. The scat collections are made in a fashion which allows each scat to be assigned to a period in time (essentially, the period since the last collection).

There were 36 collections made over a period of 5 years, from September 1981 through September 1985. Of these 36 collections, all but 2 each contained more than 10 and generally several 10's of scats (12 to 95). By pooling 1 of the sparse collections with another that was close to it in time, and discarding the remaining sparse collection, an effective time series of 34 collections, each containing more than 10 and less than 100 scats, was obtained.

The time series of 34 effective collections does not represent uniform intervals. Most collections were taken at intervals of approximately 1 or 2 months, but there were three gaps of longer duration (3 or 4 months approximately). Those gaps were winter of 1981-1982, fall of 1983 and spring of 1985.

The time series includes a period of unusual oceanographic conditions attributable to the 82-83 El Niño. The sea surface temperatures in the region were abnormally warm from October 1982 through November 1984.

The entire time series represents 1,347 scats, of which 1,190 contained identifiable prey remains. The 157 scats containing no identifiable taxa represent genuine zeros, and contribute to the effective sample size. Individual scats contained zero to more than 5 taxa, and zero to (evidently) more than 100 individual counts of a given taxon in a given scat.

II. NOTES ON THE NATURE OF THE DATA

(a) Quasi-integrating sampling

The scats were collected only at intervals (generally of a few months). The areas where scats were collected were cleared of scats at the time of each collection, so that scats obtained at the next collection may be assumed to have been produced in the interval. To this extent, then, the collection at a particular visit to the rookery represents an integrating sample of the scats produced since the last visit--as distinguished from a sample pertaining only to a point in time, with "gaps" in the record between every pair of visits.

I qualify the characterization of these samples as integrating samples, because we don't know what mechanisms of attrition might cause the sample obtained on a given visit to be more representative of the most recent portion of the interval since the last visit, owing to loss of a fraction of the scats that are representative of the earliest (oldest) portion of the interval. I can imagine the following may occur: disintegration owing to weathering, removal or fracturing by scavengers, and loss owing to random burial. The important question is how fast these processes operate, compared to the collection interval (say 2 or 3 months). To this end, I recommend some field experiments to determine the "turnover rate" of scats. Most simply, clear an area, and census the scats in that area (without removing them) at closer intervals (e.g. 1/2 month) to see how long it takes before the natural rate of attrition begins to affect the rate of accumulation of scats. A more elegant experiment would be to "mark" scats (without removal), for example with a salmonid wire nose tag which is detectable magnetically, to determine the "mortality rate" of these marked scats and the "birth rate" of unmarked scats.

I will presume, for the moment that the turnover rate really is low enough to treat the collections as more-or-less integrating samples, but this needs to be pursued empirically.

(b) Quasi-absolute counts

To the extent that the contents of each scat are enumerated exhaustively, the counts constitute an actual absolute count--as distinguished from data reflecting relative (percent) composition only.

On the other hand, the data do not tell us how many sea lions were responsible for the collected scats, or how many of the scats by the responsible sea lions were not collected owing to their being deposited outside the collection area. Thus, we do not have the data to express the diet in terms of prey items per individual sea lion per unit time. Functionally, the sea lions eat at whatever rate they eat, and produce a scat whenever a certain volume of food has been processed, so that the scat, as a sampling unit represents a package of a purely arbitrary, more-or-less fixed, size, and to this extent the counts in a scat represent only a sample reflecting relative prey composition in the diet (even though they are absolute counts with respect to the scat).

For certain purposes, there will be a gain in power of the analysis by treating the scats as the fundamental unit, and treating the counts in a scat as absolute. But it must be borne in mind that these absolute counts pertain to the scat, and not strictly to "the diet." With respect to the diet, in the broadest sense, the counts are essentially relative.

A combination of these two perspectives arises when we consider that the scat functionally represents a discrete (and consecutive) number of feeding episodes, so to this extent the absolute counts in the scat do genuinely represent a functional subdivision of "the diet."

III. THE SAMPLE-SIZE QUESTION

These comments were initially prompted by the question: "What sample size is necessary to detect a substantial change in diet?" By this we mean: If we in some way pool the counts from a set of scats (several scats, presumably from several individual sea lions, from several collection periods) and then compare some summary statistic (e.g. % comprised by a particular prey taxon), to the same statistic from a second set, how can we be reasonably sure that the difference between the two sets is not due to random sampling?

In order to address the question of sample size, we need first to define the nature of an "event" that we would wish to detect with some defined level of security. This involves defining quantitatively the duration of the event, the intensity of the event, the sampling schedule, and the desired certainty. I will develop an example here, but it must be understood that some of the decisions are arbitrary, and may be revised to suit the questions that are being asked of the data.

Inspection of the graphs of percent occurrence (and percent minimum number) against time in the reports by Lowry & Oliver (1986) and Lowry, Oliver and Wexler (1986), suggests that "major events" in the frequency in the diet of the most common prey items take the form of episodes of about 1 year duration where the percent occurrence is roughly 3 to 5 times greater than at most other times. Given that the samples are more-or-less integrating, I judge that a collection interval of 2 months (sampling every other month) is adequate to resolve the gross features of the onset and termination of a major event. Let us assume that the goal is to determine, with this temporal resolution, when a major event is taking place.

(a) Events in the fraction of scats containing a given prey taxon

We may define the event in terms of the fraction of scats containing the subject prey taxon exceeding a value three times its usual value. On the assumption that a usual event will have a duration of about a year, and that samples are taken every other month, the event should be detectable in 6 consecutive collections. Let us say that if the fraction containing that taxon fell below half the "event threshold" (e.g. below 1.5 times the usual fraction), we would decide that the event was over (or hadn't begun yet, or was in remission). Then our objective might be to set a sample size for which, the probability was 5% or less that none of the 6 collections (which sampled scats where the true fraction occurrence was 3 times the usual value) would exhibit a fraction occurrence lower than 1.5 times usual. A major prey taxon might be one which was usually present in about 15% of the scats.

To formalize our objective, we have

$m = 6$ = the number of collections for which the fraction occurrence is to be within the desired bounds in each collection.

$x = 0.15$ = the usual fraction occurrence (this can be the mean, mode, or any other meaningful expression of central tendency--it serves only as a reference value against which measures of departure from usual are scaled, as in the parameters h and l, defined below).

$h = 3.0$ = the factor by which the true fraction occurrence must increase over usual to catch our attention as a major event.

$l = 1.5$ = the factor by which the fraction occurrence in a sample must increase over usual to catch our attention as indicating an event.

a = 0.05 = our tolerance for false negatives during the course of a six-collection event.

We presume initially that a sample size which satisfies this objective will probably prove acceptable with respect to the frequency of false positives (i.e. the frequency with which samples, during a time of usual true fraction occurrence, exhibit a sample fraction occurrence greater than 1.5 times the usual value); we may confirm this by computing the probability of a single such false positive (which I suspect does not concern us much) or the frequency of a consecutive pair of false positives (which probably would be more important).

Where the objective is a probability of (1.0 - a) or more for m consecutive collections, during a true event, not showing a sample false negative, the acceptable probability for any one collection showing a false negative, assuming independence between collections, is

$$b = 1.0 - (1.0 - a)^{(1/m)} \qquad [1]$$

Because we are dealing with presence or absence only, where the scat is the sampling unit, we can conveniently work from an assumption of a binomial sampling distribution for the fraction of scats containing a given prey taxon in a given collection. Where the sample size, per collection, is n, and the true fraction of scats containing the prey is h*x, the variance in the number of scats in the sample containing the prey taxon is n*h*x(1.0 - h*x), and the mean number of scats in the sample containing the prey taxon is n*h*x. For the fraction of scats in the sample containing the prey, the mean is h*x and the variance is h*x(1.0-h*x)/n.

Approximating the binomial as a gaussian with the same mean and variance, our criterion is to find a value for n so that the tail of the distribution below the threshold 1*h*x represents a probability b. Standardizing the distribution, by subtracting the mean h*x and dividing by the standard deviation, the threshold becomes

$$t = x(1-h)/\mathrm{SQRT}(h*x(1.0-h*x)/n) \qquad [2]$$

A table of areas of the standard normal distribution will give us the deviate z corresponding to a lower tail representing the area b. In Table P of Rohlf and Sokal, for example, we would look up the negative of the value of z corresponding to the area (1/2 -b).

Setting t of eq [2] equal to the value of z corresponding to the threshold for a lower tail of area b from eq [1], we could then solve for n:

$$n = h\ (1/x - h)\ (z/(1-h))^{2} \qquad [3]$$

Given this sample size, the distribution of the fraction of scats in the sample containing the prey taxon, when the true fraction was x, would be a binomial with mean x and variance x(1.0-x)/n. Again approximating this as a guassian, and standardizing, the probability of a false positive in a collection during a time of true frequency x, when we score a positive if the sample fraction equals or exceeds 1*x, would be found from the area of the upper tail beyond the threshold x(1-1.0)/SQRT(x(1.0-x)/n) of a standard normal.

With the above suggested values, the calculated required sample size is about 28 scats on each collection date, and the associated probability, p, of a false positive is about 0.13 in one collection, and about 0.02 for a pair of consecutive collections, confirming the false negatives results in a rate of false positives which seems acceptable.

The following table shows the consequences of systematically varying the specifications for each of the parameters in turn:

m	a	x	h	l	n	p	p**2
6	.05	.15	3.0	1.5	27.8	0.134	0.018
5	.05	.15	3.0	1.5	26.3	0.141	0.020
6	.10	.15	3.0	1.5	21.8	0.163	0.027
6	.05	.05	3.0	1.5	129.1	0.096	0.009
6	.05	.15	4.0	1.5	9.7	0.256	0.066
6	.05	.15	3.0	2.0	62.7	0.000	0.000

Where:

m is the number of collections for which the fraction occurrence is to be within the desired bounds in each collection.

a is our tolerance for false negatives during the course of a six-collection event.

x is the usual fraction occurrence.

h is the factor by which the true fraction occurrence must increase over usual to catch our attention as a major event.

l is the factor by which the fraction occurrence in a sample must increase over usual to catch our attention as indicating an event.

n is the required minimum sample size.

p is the resulting probability of a false positive in a given collection.

p**2 is the resulting probability of 2 consecutive false positives.

In order that the user may explore the consequences of other parameter values, an interactive Fortran program is enclosed (Appendix 1) which accepts as input the parameters (m,x,h,l) and computes (n,p,p**2). This program can be invoked by typing SCAT with the disk logged on a CPM computer. If it is desired to modify the program to run on other equipment, a complete listing of the Fortran is appended.

(b) Events in the mean number of prey individuals per scat

Simply analyzing the fraction of the scats containing a given prey taxon, treats that data as presence/absence only, neglecting the information content of the counts themselves. In order to use this information, we might wish to define events in terms of episodes when the mean number of individuals per scat, of a given prey taxon, was unusually high.

Conceivably, the results would be similar to the analysis of fraction occurrence, since inspection of the graphs in Lowry and Oliver (1986), and in Lowry, Oliver and Wexler (1986), indicate that the indices "% occurrence" and "% minimum number" showed very similar time trajectories for the common prey items.

The key point to considering the number of prey individuals per scat is that the distribution of these numbers may be non-random. During a given feeding episode, an individual sea lion may "specialize" in feeding on particular items in particular places, so that the corresponding scat may reflect a very different prey composition than a scat from a different sea lion at the same time or a scat from the same sea lion at a different time. Depending on the distribution of these numbers, the variance in prey numbers among scats can be very large, or quite small, for the same value of the presence/absence rate of this prey taxon.

It would behoove us initially to test specifically for randomness. The null hypothesis would be that the distribution of the number of individuals of prey (of a given taxon) per scat is random. Under this null hypothesis, the expected distribution is Poisson, so the natural thing to do is compare the observed frequency of scats with 0, 1, 2, 3, etc. individuals of the subject prey taxon against a Poisson distribution with the same mean number of prey per scat as observed. A chi-square test will serve to test the significance of the fit.

If the null hypothesis is accepted, there is no evidence of "specialization" in feeding episodes, so a Poisson model (where the variance equals the mean) will suffice to describe the sampling variance. Under this null model, we would expect the "% minimum number" index to mirror closely the "% occurrence" index, so the calculations of Section III (a) would probably suffice for estimating a minimum sample size for purposes of considering number of prey individuals per scat.

If the null hypothesis is rejected, there is evidence of some sort "specialization," or strongly uniform "selectivity", and accordingly we must use some more elaborate means for estimating the sampling variance in our estimates of the mean number of prey individuals per scat. A straightforward approach would be to bootstrap subsampling of a collection from a single date when there was a particularly large number of scats collected. This would then yield a variance estimate which would be used in place of the variance based on a binomial in a calculation like that described above under (a) "Events in the fraction of scats...." Alternatively, we might compute the variance in such a collection, and continue to assume normality of the sampling distribution.

IV. USES FOR THE DATA

(1) Monitoring

A monitoring program, using a collection sample size informed by the calculations of Section III will provide a satisfactory basis for detecting changes of a magnitude no smaller than specified in the parameters for the calculation. For practical reasons, this will usually mean that large diet shifts are detectable.

We are of course tempted to seek an "effect" of the 82-83 El Niño in the existing data. I suspect that the data are not yet sufficient to resolve such an effect. The El Niño lasted a year; there are now but 4 years data, and inspection of the data suggests an inherent serial correlation (temporal patchiness) on the order of a year. In other words, the diet undergoes major changes in blocks of time of about a year, and we only have an effective sample of 4 such blocks. Probably a 10-year monitoring record must be accumulated to build a convincing case about El Niño.

(2) Components of variability

The existing data do lend themselves to an analysis of variance, partitioning the variation into: year effects, season effects, and sampling variation between scats. I suspect that the year effect will prove to be overwhelming, for most of the common prey taxa; but it is worth documenting this quantitatively.

(3) Details of the feeding strategy

The data set contains a wealth of information that could be analysed to reveal details of the way the sea lions conduct their feeding. The description of the "specialization" issue under Section III (b) introduces one aspect of feeding strategy. Other aspects worth pursuing include 2 in particular: Patterns of co-occurrence of certain prey types (especially as they might be consistent with temporal or spatial patterns of ambient abundance of the respective taxa), and patterns in the portion of the diet made up of rare versus common taxa.

These matters are worthy of further analysis: they are biologically very interesting, and may have implications for management, but of course though they fall outside the original scope of the question of "sample size" which is the focus of this report.

LITERATURE CITED

Lowry, M.S. and C.W. Oliver. 1986. The food habits of the California sea lion, Zalophus californianus, at San Clemente Island, California, September 1981 through March 1983. NOAA/SWFC Admin. Rept. LJ-86-07.

Lowry, M.S., C.W. Oliver, and J.B. Wexler. 1986. The food habits of California sea lions at San Clemente Island, California: April 1983 through September 1985. NOAA/SWFC Admin. Rept. LJ-86-33.

APPENDIX 1

PROGRAM SCAT.FOR

```
      PROGRAM SCAT
C*****COMPUTES MINIMUM SAMPLE SIZE FOR SCAT COLLECTION
C     WHERE INTEREST IS IN FRACTION OF SCATS CONTAINING THE
C     PREY ITEM.
C     approximates THE BINOMIAL WITH A GAUSSIAN.
C     -------------------------------------------------------------
C     |  Written in standard 1966 Fortran. All WRITEs are to the  |
C     |  CRT screen, and all READs are from the terminal keyboard.|
C     |  The device numbers for these IO operations are set in    |
C     |  the initial DATA statement. These device numbers may     |
C     |  need to be altered for installation on various hardware. |
C     |  A very few Fortrans will not digest the PROGRAM statement|
C     |  which must be deleted for such installations. Otherwise, |
C     |  the code should be 100% portable; and it should recompile|
C     |  on any Fortran compiler.                                 |
C     -------------------------------------------------------------
C
C*****
C     VERSION OF JUNE 27, 1986
C*****
      DATA NCRT/3/,NCON/1/
 1000 WRITE (NCRT,9010)
 9010 FORMAT (1X,'SAMPLE SIZE ANALYSIS FOR FOOD HABITS STUDY'/
     *1X,'Deals with fraction of scats containing the given'
     *1X,'prey taxon.'/
     *1X,'(D. Goodman, 6/27/86)')
      WRITE (NCRT,9020)
 9020 FORMAT (/
     *1X,'Specify M, the number of successive collections which we'/
     *1X,'want to be free of false negatives.'/
     *1X,'M=? ')
      READ (NCON,9030) M
 9030 FORMAT (I5)
      WRITE (NCRT,9040)
 9040 FORMAT (
     *1X,'Specify the tolerable frequency (as a fraction, not a'/
     *1X,'percent) for mistaken determination of time of onset or'/
     *1X,'decline in an "episode" of increased utilization of this'/
     *1X,'prey taxon.'/
     *1X,'A=? ')
      READ (NCON,9050) A
 9050 FORMAT (E17.4)
      WRITE (NCRT,9060)
 9060 FORMAT
     *(1X,'Specify the usual occurrence (as a fraction of scats) of'/
     *1X,'the prey taxon in question.'/
     *1X,'X=? ')
      READ (NCON,9050) X
      WRITE (NCRT,9070)
```

```
 9070 FORMAT (
     *1X,'Specify the multiplicative factor by which the true'/
     *1X,'occurrence increases during an "event" (eg. 3.0 means the'/
     *1X,'occurrence rate increases three-fold). Be sure to put'/
     *1X,'a decimal point in this number.'/
     *1X,'H=? ')
      READ (NCON,9050) H
      WRITE (NCRT,9080)
 9080 FORMAT (
     *1X,'Specify the multiplicative factor by which the observed'/
     *1X,'occurrence rate in the sample must increase above the usual'/
     *1X,'value for the sample to be considered indicative of an'/
     *1X,'"event." Be sure to put a decimal point in this number also.'/
     *1X,'L=? ')
      READ (NCON,9050) FL
      WRITE (NCRT,9090) M,A,X,H,FL
 9090 FORMAT (1X,'For parameter values:'/1X,'M= ',I3/
     *1X,'A= ',F7.3/1X,'X= ',F8.4/1X,'H= ',F5.2/1X,'L= ',F5.2)
C.....CARRY OUT SAMPLE SIZE CALCULATION
      B=1.0-(1.0-A)**(1.0/FLOAT(M))
      CALL NRINV(B,Z)
      FN=(Z/(H-FL))**2
      FN=FN*H*(1.0/X-H)
C.....CARRY OUT FALSE POSITIVE CALCULATION
      ZN=SQRT(FN*X/(1.0-X))*(FL-1.0)
      CALL NORPR(ZN,P)
      P=P/2.0
      P2=P*P
C.....OUTPUT
      WRITE (NCRT,9100) FN
 9100 FORMAT (1X,'Minimum number of scats per collection=',F12.1)
      WRITE (NCRT,9110) P
 9110 FORMAT (/
     *1X,'Probability, with this sample size, that any given'/
     *1X,'collection, during a "non-event", will yield a false'/
     *1X,'positive= ,                          ',F12.6)
      WRITE (NCRT,9120) P2
 9120 FORMAT (
     *1X,'Probability that two consecutive samples are false'/
     *1X,'positive=                            ',F12.6)
      WRITE (NCRT,9150)
 9150 FORMAT(///1X,'Enter 1 to begin a new calculation, 0 to exit.')
      READ (NCON,9030) K
      IF (K) 9999,9999,1000
C.....
 9999 CALL EXIT
      END
      SUBROUTINE NRINV(P,X)
C*****INVERSE OF CUMULATIVE DISTRIBUTION FUNCTION FOR GAUSSIAN WITH
C     ZERO MEAN AND UNIT VARIANCE.
C.....USES HASTINGS APPROXIMATION FOR THE FUNCTION
C                   X=G(P)
C     WHERE         P=F(X)
C     IS THE INTEGRAL OF THE NORMAL CURVE FROM MINUS INFINITY TO X.
C.....NOTE THAT VALUES TOO NEAR (AND INCLUDING) ZERO OR ONE FOR P
```

```
C     WILL CAUSE AN ILLEGAL ARGUMENT CONDITION (LOG OF ZERO) AT
C     STATEMENT 1070. THESE CORRESPOND TO X OF PLUS OR MINUS INFINITY.
C.....ONCE THE MACHINE SPECIFIC VALUES ARE FOUND, THEY COULD BE
C     TRAPPED WITH AN IF STATEMENT AT 1070.
C*****
C     VERSION OF AUG 23, 1981.
C*****
      DATA AZ/2.515517/,A1/0.802853/,A2/0.010328/,
     *     B1/1.432788/,B2/0.189269/,B3/0.001308/
C.....
      PM5=P-0.5
      IF (PM5) 1060,1050,1040
 1040 Q=1.0-P
      GO TO 1070
 1050 X=0.0
      GO TO 1090
 1060 Q=P
 1070 T=SQRT(-2.0*ALOG(Q))
      T2=T*T
      T3=T2*T
      X=T-(AZ+A1*T+A2*T2)/(1.0+B1*T+B2*T2+B3*T3)
      IF (PM5) 1080,1050,1090
 1080 X=-X
C.....
 1090 RETURN
      END
      SUBROUTINE NORPR(Z,PR)
C*****RETURNS THE TWO-TAILED PROBABILITY OF FALLING AT LEAST Z
C     STANDARD DEVIATIONS FROM THE MEAN IN A NORMAL DISTRIBUTION.
C.....THIS VERSION USES HASTING'S APPROXIMATION FOR THE INTEGRAL
C     OF A NORMAL PROBABILITY DENSITY FUNCTION.
C.....RUN IN SINGLE PRECISION, THE VALUES AGREE WITH THE TABLES
C     IN SOKAL AND ROHLF OUT TO THE FOURTH DIGIT PAST THE DECIMAL
C     POINT.
C*****
C     VERSION OF FEB 18, 1981.
C*****
      DATA A/0.79788456/,P/0.2316419/,B1/0.319381530/,B2/-0.356563782/,
     *     B3/1.781477937/,B4/-1.821255978/,B5/1.330274429/
C.....
      T=1.0/(1.0+P*ABS(Z))
      PR=A*EXP(-Z*Z/2.0)*T*(B1+T*(B2+T*(B3+T*(B4+T*B5))))
C.....
      RETURN
      END
```